Solution to Homework #7 (Seongwon Kang)

momentum eq.

$$\rho\left(u\frac{\partial u}{\partial x}+v\frac{\partial u}{\partial y}\right)=\rho_e U_e\frac{dU_e}{dx}+\frac{\partial}{\partial y}\left(\mu\frac{\partial u}{\partial y}\right),$$

$$\frac{\partial(\rho u)}{\partial x}+\frac{\partial(\rho v)}{\partial y}=0,$$

B.C) $u(x,0)=v(x,0)=0$

$u(x,\infty)=U_e=U_0(1+x/4L)$

I.C) $u(0,y)=U_{uniform}$

$v(0,y)=0$

Let's nondimensionalize using following variables

$$\rho^*=\frac{\rho}{\rho_0},\quad x^*=\frac{x}{L},\quad y^*=\frac{y}{\delta},\quad u^*=\frac{u}{U_0}$$

$$v^*=\frac{v}{U_0}\frac{L}{\delta},\quad \mu^*=\frac{\mu}{\mu_0}=\frac{\mu}{\frac{\rho_0 U_0\delta^2}{L}}$$

Then

$$\begin{cases}\rho^*\left(u^*\frac{\partial u^*}{\partial x^*}+v^*\frac{\partial u^*}{\partial y^*}\right)=\rho_e^* U_e^*\frac{dU_e^*}{dx^*}+\frac{\partial}{\partial y^*}\left(\mu^*\frac{\partial u^*}{\partial y^*}\right)\\ \frac{\partial(\rho^* u^*)}{\partial x^*}+\frac{\partial(\rho^* v^*)}{\partial y^*}=0\end{cases}$$

Using Busemann energy integral

$$h_0 = h + \frac{u^2}{2} = \text{const.}$$

$$\Rightarrow \underline{h^* + \frac{u^{*2}}{2} = \text{const.}}, \text{ where } h^* = \frac{h}{U_0^2}, u^* = \frac{u}{U_0}$$

From $h = C_p T$

$$\Rightarrow \underline{h^*} = \frac{C_p T_0}{U_0^2} T^* \qquad C_p = \frac{\gamma R}{\gamma - 1}$$

$$= \frac{\gamma R T_0}{U_0^2} \frac{1}{\gamma-1} T^*$$

$$= \underline{\frac{1}{Ma^2} \frac{1}{\gamma - 1} T^*}$$

From $p = \rho R T$

assuming $p(y) \approx p_e$ since BL is thin,

$$\frac{\rho}{\rho_e} = \frac{T_e}{T} \Rightarrow \underline{\frac{\rho^*}{\rho_e^*} = \frac{T_e^*}{T^*}}$$

From constant entropy s assumption

$$T \cancel{ds} = dh - v\,dp$$

$$\Rightarrow C_p\,dT = \frac{1}{\rho} dp$$

$$= R\,dT + \frac{R}{\rho} T\,d\rho$$

$$\Rightarrow C_v \frac{dT}{T} = R \frac{d\rho}{\rho} \Rightarrow T = C \rho^{\frac{R}{C_v}}$$

Note that the constant entropy assumption is only valid in the free-stream. In the boundary layer, pressure and density are obtained from the free-stream quantities using the ideal gas law and the fact that $p = p_e$.

$$\therefore \quad \frac{T_1}{T_2} = \left(\frac{\rho_1}{\rho_2}\right)^{R/c_v} = \left(\frac{\rho_1}{\rho_2}\right)^{\gamma-1}$$

$$\Rightarrow \quad \underline{T^* = \rho^{*\,\gamma-1}}$$

From $\underline{\mu = \beta T^n}$

$$\therefore \quad \frac{\mu}{\mu_0} = \frac{\beta T^n}{\beta T_0^n} \Rightarrow \underline{\mu^* = T^{*n}}$$

where $\mu_0 = \beta T_0^n = \frac{\rho_0 U_0 \delta^2}{L}$

<u>In summary,</u>

Compute
$$\begin{cases} h^* \text{ from } h^* + \frac{u^{*2}}{2} = h_0^* \\ T^* \text{ from } h^* = \frac{1}{M_a^2}\frac{1}{\gamma-1} T^* \\ \rho^* \text{ from } \rho^* = T^{*\frac{1}{\gamma-1}} \ (\text{for } \rho^*_e) \text{ ok.} \\ \quad \text{from } \frac{\rho^*}{\rho_e^*} = \frac{T_e^*}{T^*} \\ \mu^* \text{ from } \mu^* = T^{*n} \end{cases}$$

Using <u>implicit Euler method</u>, from White

$$-\alpha^* U_{m+1,n+1} + (1+2\alpha^*) U_{m+1,n} - \alpha^* U_{m+1,n-1}$$

$$= U_{m,n} + \frac{\rho_{e,m}(U_{e,m+1}^2 - U_{e,m}^2)}{2\rho_{mn} U_{m,n}} - (\beta - \gamma)(U_{m,n+1} - U_{m,n-1})$$

where

$$\alpha^* = \frac{\mu_{m,n}\Delta x}{\rho_{m,n} U_{m,n} \Delta y^2}, \quad \beta = \frac{V_{m,n}\Delta x}{2U_{m,n}\Delta y}, \quad \gamma = \frac{(\mu_{m,n+1} - \mu_{m,n-1})\Delta x}{4\rho_{m,n} U_{m,n}\Delta y^2}$$

Computational procedure

1) From U_e, compute T_e, ρ_e, μ_e

2) Marching in x using discretized momentum eq. to get u at $m+1$

3) compute $T_{m+1}, \rho_{m+1}, \mu_{m+1}$

4) compute V_{m+1} using continuity eq

Test case : Blasius case $U_e = U_0 = \text{const.}$

$M_a = 1.5$, $\gamma = 1.4$ for air, $n = 1$

for nondimensionlization, following relation is used

Corrected mistake in Seongwon's solution (3/10/04): $x^* = x/L$, $v^* = (v/U_0)(L/\delta)$

$$x^* = \frac{x}{L}, \quad y^* = \frac{y}{\delta}, \quad u^* = \frac{u}{U_0}, \quad v^* = \frac{v}{U_0}\frac{L}{\delta}$$

Domain size in x : $L_x = \underline{2\delta}$

in y : $L_y = \underline{4\delta}$

grid points in x : $N_x = \underline{800}$

in y : $N_y = \underline{800}$





Profiels are developing as downstream.

Main case

$$U_e = U_0 \left(1 + \frac{x}{4L}\right) \Rightarrow U_e^* = 1 + \frac{x^*}{4}$$

$M_a = 1.5$, $\gamma = 1.4$, $n = 2/3$

Domain size in x : $L_x = \underline{4L}$

in y : $L_y = \underline{8\delta}$

\# grid points in x : $N_x = \underline{1200}$

$N_y = \underline{1200}$

In this case, U_e is developing as downstream and at certain x^*, T^* will become negative from $h^* + \frac{U^{*2}}{2} = \text{const}$.

then further marching will be impossible





Unlike Blasius case, farfield velocity increases and farfield temperature decreases as downstream.

[redacted]

```
C========================================================================
C  FORTRAN CODE FOR COMPRESSIBLE BOUNDARY LAYER FLOW
C  FOR ME 351B
C                                              MADE BY SEONGWON KANG
C========================================================================
      PROGRAM MAIN

      PARAMETER(M_X=1500)
      PARAMETER(M_Y=1300)

      REAL X(0:M_X+1),Y(0:M_Y+1)
      REAL U(0:M_X+1,0:M_Y+1),V(0:M_X+1,0:M_Y+1)
      REAL T(0:M_X+1,0:M_Y+1),RHO(0:M_X+1,0:M_Y+1),VSC(0:M_X+1,0:M_Y+1)
      REAL, DIMENSION (M_Y) :: AU,BU,CU,RU,UU

      INTEGER IPOS(6)
      CHARACTER*20 FNAME(6)

      ! INPUT PARAMETERS
      ! ----------------
      X_L=4.
      Y_L=8.
      N_X=1199
      N_Y=1199
      ! ----------------

      ! INITIAILIZATION OF GRID
      ! ----------------
      DX=X_L/(N_X+1)
      DY=Y_L/(N_Y+1)

      DO I=0,N_X+1
       X(I)=I*DX
      END DO
      DO J=0,N_Y+1
       Y(J)=J*DY
      END DO

      JE=N_Y+1
      ! ----------------

      ! INITIAIL & BOUNDARY VELOCITY
      ! ----------------
      U=0.; V=0.

c      DO I=0,N_X+1         ! BLASIUS
c       U(I,JE)=1.
c      END DO

      DO I=0,N_X+1         ! ACCELERATING FLOW
       U(I,JE)=1+X(I)/4
      END DO

      DO J=0,N_Y
       U(0,J)=U(0,JE)
```


[redacted]

```
       V(0,J)=0.
      END DO
      ! ----------------

      ! COMPRESSIBLE PROPERTIES
      ! ----------------------
      GAMMA=1.4
      V_MACH=1.5

c      POWER=1.       ! BLASIUS
      POWER=2./3.     ! ACCELERATING FLOW

      T_INF=1.
      H_STAG=T_INF/( V_MACH**2*(GAMMA-1) )
      ! ----------------------

      ! INITIAIL & BOUNDARY COMPRESSIBLE PROPERTIES
      ! ----------------------
      DO I=0,N_X+1
       T(I,JE)=(H_STAG-0.5*U(I,JE)**2)*V_MACH**2*(GAMMA-1)
       RHO(I,JE)=T(I,JE)**(1./(GAMMA-1))
      END DO

      DO J=0,N_Y+1
       T(0,J)=(H_STAG-0.5*U(0,J)**2)*V_MACH**2*(GAMMA-1)
       RHO(0,J)=RHO(0,JE)*T(0,JE)/T(0,J)
       VSC(0,J)=T(0,J)**POWER
      END DO
      ! ----------------------


      ! MARCHING IN X
      ! ==============================
      DO I=1,N_X
       IM=I-1

       ! COMPUTATION OF U
       DO J=1,N_Y
        ALP=VSC(IM,J)*DX/( RHO(IM,J)*U(IM,J)*DY**2 )
        BET=V(IM,J)*DX/( 2*U(IM,J)*DY )
        SIG=(VSC(IM,J+1)-VSC(IM,J-1))*DX/
     &       ( 4*RHO(IM,J)*U(IM,J)*DY**2 )

        AU(J)=-ALP
        CU(J)=-ALP
        BU(J)=1+2*ALP
        RU(J)=U(IM,J)+RHO(IM,JE)*(U(I,JE)**2-U(IM,JE)**2)/
     &        ( 2*RHO(IM,J)*U(IM,J) )-
     &        (BET-SIG)*(U(IM,J+1)-U(IM,J-1))
       END DO
       RU(1)=RU(1)-AU(1)*U(I,0)
       RU(N_Y)=RU(N_Y)-CU(N_Y)*U(I,JE)

       CALL TRDIAG_A(M_Y,1,AU,BU,CU,RU,UU,1,N_Y,1,1)
```

```
      DO J=1,N_Y
       U(I,J)=UU(J)
      END DO

      ! COMPUTATION OF T,RHO,VSC
      DO J=0,N_Y+1
       T(I,J)=(H_STAG-0.5*U(I,J)**2)*V_MACH**2*(GAMMA-1)
       RHO(I,J)=RHO(I,JE)*T(I,JE)/T(I,J)
       VSC(I,J)=T(I,J)**POWER
      END DO

      ! COMPUTATION OF V
      DO J=1,N_Y+1
       V(I,J)=RHO(I,J-1)*V(I,J-1)/RHO(I,J)-DY/(2*RHO(I,J)*DX)*
     &         ( RHO(I,J)*U(I,J)-RHO(IM,J)*U(IM,J)+
     &           RHO(I,J-1)*U(I,J-1)-RHO(IM,J-1)*U(IM,J-1)  )
      END DO

     END DO
     ! ================================


     ! OUTPUT
     ! -------------
     IPOS(1)=20   ; FNAME(1)='v1'
     IPOS(2)=60   ; FNAME(2)='v2'
     IPOS(3)=100  ; FNAME(3)='v3'
     IPOS(4)=140  ; FNAME(4)='v4'

     DO II=1,4
      OPEN(3,FILE=FNAME(II))
      DO J=0,N_Y+1
       WRITE(3,88)Y(J),U(IPOS(II),J),V(IPOS(II),J),T(IPOS(II),J)
      END DO
      CLOSE(3)
     END DO


  88 FORMAT(4E16.8)

     STOP
     END


C********************** TRDIAG_A *******************************
C     SOLVE TRIDIAGONAL MATRICES WITH
C     DIFFERENT COEFS.

      SUBROUTINE TRDIAG_A(M1,M2,A,B,C,R,UU,L1,L2,LL1,LL2)

      REAL, DIMENSION(M2,M1) :: A,B,C,R,UU,GAM
      REAL, DIMENSION(M2) :: BET

      DO I=LL1,LL2
       BET(I)=B(I,L1)
```



```
       UU(I,L1)=R(I,L1)/BET(I)
      END DO

      DO J=L1+1,L2
      DO I=LL1,LL2
       GAM(I,J)=C(I,J-1)/BET(I)
       BET(I)=B(I,J)-A(I,J)*GAM(I,J)
       UU(I,J)=(R(I,J)-A(I,J)*UU(I,J-1))/BET(I)
      END DO
      END DO
      DO J=L2-1,L1,-1
      DO I=LL1,LL2
       UU(I,J)=UU(I,J)-GAM(I,J+1)*UU(I,J+1)
      END DO
      END DO

      RETURN
      END
```